

# **Elmer Programmer's Tutorial**

Mikko Lyly

CSC – IT Center for Science

Spring 2010

# Elmer Programmer's Tutorial

## About this document

The Elmer Programmer's Tutorials is part of the documentation of Elmer finite element software. It gives examples on how to carry out simple coding tasks using the high-level routines from Elmer library.

The present manual corresponds to Elmer software version 6.2. Latest documentations and program versions of Elmer are available (or links are provided) at <http://www.csc.fi/elmer>.

## Copyright information

The original copyright of this document belongs to CSC – IT Center for Science, Finland, 1995–2009. This document is licensed under the Creative Commons Attribution-No Derivative Works 3.0 License. To view a copy of this license, visit <http://creativecommons.org/licenses/by-nd/3.0/>.

Elmer program is free software; you can redistribute it and/or modify it under the terms of the GNU General Public License as published by the Free Software Foundation; either version 2 of the License, or (at your option) any later version. Elmer software is distributed in the hope that it will be useful, but without any warranty. See the GNU General Public License for more details.

Elmer includes a number of libraries licensed also under free licensing schemes compatible with the GPL license. For their details see the copyright notices in the source files.

All information and specifications given in this document have been carefully prepared by the best efforts of CSC, and are believed to be true and accurate as of time writing. CSC assumes no responsibility or liability on any errors or inaccuracies in Elmer software or documentation. CSC reserves the right to modify Elmer software and documentation without notice.

# Contents

<b>Table of Contents</b>	<b>2</b>
<b>1 User defined functions</b>	<b>3</b>
1.1 Calling convention . . . . .	3
1.2 Compilation . . . . .	4
<b>2 User defined solvers</b>	<b>5</b>
2.1 Calling convention . . . . .	5
2.2 Compilation . . . . .	5
2.3 Solver Input File . . . . .	5
2.4 Exercises . . . . .	5
<b>3 Reading constant data from SIF</b>	<b>7</b>
3.1 Reading constant scalars . . . . .	7
3.2 Reading constant vectors . . . . .	7
3.3 Reading constant matrices . . . . .	8
3.4 Exercises . . . . .	8
<b>4 Partial Differential Equations</b>	<b>10</b>
4.1 Model problem . . . . .	10
4.2 FEM . . . . .	10
4.3 Implementation . . . . .	11

# Chapter 1

## User defined functions

### 1.1 Calling convention

All user defined functions that implement e.g. a material parameter, body force, or a boundary condition, are written in Fortran90 with the following calling convention:

```
!-----
! File: MyLibrary.f90
! Written by: ML, 5 May 2010
! Modified by: -
!-----
FUNCTION MyFunction(Model, n, f) RESULT(g)
    USE DefUtils
    TYPE(Model_t) :: Model
    INTEGER :: n
    REAL(KIND=dp) :: f, g

    ! code

END FUNCTION MyFunction
```

The function is called automatically by ElmerSolver for each node index  $n$ , when activated from the Solver Input File e.g. as follows:

```
Material 1
  MyParameter = Variable Time
  Real Procedure "MyLibrary" "MyFunction"
End
```

In this case, the value of time will be passed to the function in variable  $f$ . The function then returns the value of the material parameter in variable  $g$ .

The type `Model_t` is declared and defined in the source file `DefUtils.f90`. It contains the mesh and all model data specified in the Solver Input File. As an example, the coordinates of node  $n$  are obtained from `Model` as follows:

```
REAL(KIND=dp) :: x, y, z
x = Model % Nodes % x(n)
y = Model % Nodes % y(n)
z = Model % Nodes % z(n)
```

If the value of the return value depends on a specific function (for example a temperature dependent heat conductivity), we can fetch the nodal value of that function by using the `DefUtils-subtoutines` (more details to follow in the next section):

```
TYPE(Variable_t), POINTER :: TemperatureVariable
REAL(KIND=dp) :: NodalTemperature
INTEGER :: DofIndex
TemperatureVariable => VariableGet(Model % Variables, 'Temperature')
DofIndex = TemperatureVariable % Perm(n)
NodalTemperature = TemperatureVariable % Values(dofIndex)
! Compute heat conductivity from NodalTemperature
```

## 1.2 Compilation

The function is compiled into a shared library (Unix-like systems) or into a dll (Windows) by using the default compiler wrapper `elmerf90` (here and in the sequel, `$` stands for the command prompt of a bash shell (Unix) and `>` is input sign of the Command Prompt in Windows):

```
$ elmerf90 -o MyLibrary.so MyLibrary.f90
```

```
> elmerf90 -o MyLibrary.dll MyLibrary.f90
```

## Chapter 2

# User defined solvers

### 2.1 Calling convention

All user defined subroutines that implement a custom solver are written in Fortran90 with the following calling convention:

```
!-----  
! File: MySolver.f90  
! Written by: ML, 5 May 2010  
! Modified by: -  
!-----  
SUBROUTINE MySolver(Model, Solver, dt, Transient)  
  Use DefUtils  
  IMPLICIT NONE  
  TYPE(Solver_t) :: Solver  
  TYPE(Model_t) :: Model  
  REAL(KIND=dp) :: dt  
  LOGICAL :: Transient  
  
  ! User defined code  
  
END MySolver
```

The types `Solver_t` and `Model_t` are defined in the source file `Types.f90`.

### 2.2 Compilation

The subroutine is compiled into a shared library like a user defined function by using the compiler wrapper `elmerf90`:

```
$ elmerf90 -o MyLibrary.so MyLibrary.f90  
> elmerf90 -o MyLibrary.dll MyLibrary.f90
```

### 2.3 Solver Input File

The user defined solver is called automatically by `ElmerSolver` when an appropriate Solver-block is found from the Solver Input File:

```
Solver 1  
  Procedure = "MyLibrary" "MySolver"  
  ...  
End
```

### 2.4 Exercises

Create a temporary work directory containing the following mesh files:

```
$ less mesh.nodes
1 -1 0.0 0.0 0.0
2 -1 0.0 -1.0 0.0
3 -1 1.0 -1.0 0.0
4 -1 1.0 1.0 0.0
5 -1 -1.0 1.0 0.0
6 -1 -1.0 0.0 0.0

$ less mesh.elements
1 1 303 1 2 3
2 1 303 1 3 4
3 1 303 1 4 5
4 1 303 1 5 6

$ less mesh.boundary
1 1 1 0 202 1 2
2 1 1 0 202 2 3
3 1 2 0 202 3 4
4 2 3 0 202 4 5
5 2 4 0 202 5 6
6 2 4 0 202 6 1

$ less mesh.header
6 4 6
2
202 6
303 4
```

Then consider the following minimalistic Solver Input File:

```
$ less case.sif
Header
  Mesh DB "." "."
End

Simulation
  Simulation Type = Steady state
  Steady State Max Iterations = 1
  Post File = case.ep
End

Body 1
  Equation = 1
End

Equation 1
  Active Solvers(1) = 1
End

Solver 1
  Equation = "MyEquation"
  Procedure = "MyLibrary" "MySolver"
  Variable = -dofs 1 "MyScalar"
End
```

Finally, make sure that your work directory contains the following info file:

```
$ less ELMERSOLVER_STARTINFO
case.sif
1
```

Write and compile a user defined subroutine that simply prints out "Hello!" when called by ElmerSolver:

```
$ ElmerSolver
ELMER SOLVER (v 5.5.0) STARTED AT: 2010/05/24 10:17:10
MAIN:
MAIN: =====
MAIN:  E L M E R   S O L V E R   S T A R T I N G
MAIN:  Library version: 5.5.0 (Rev: 4455)
MAIN: =====
MAIN:
...
Hello!
...
WriteToPost: Saving results in ElmerPost format to file ./case.ep
ElmerSolver: *** Elmer Solver: ALL DONE ***
ElmerSolver: The end
SOLVER TOTAL TIME(CPU,REAL):          0.11      0.38
ELMER SOLVER FINISHED AT: 2010/05/24 10:17:10
```

## Chapter 3

# Reading constant data from SIF

Relevant functions and subroutines (defined in DefUtils.f90):

```
RECURSIVE FUNCTION GetConstReal(List, Name, Found) RESULT(Value)
  TYPE(ValueList_t), POINTER : List
  CHARACTER(LEN=*) :: Name
  LOGICAL, OPTIONAL :: Found
  REAL(KIND=dp) :: Value

RECURSIVE SUBROUTINE GetConstRealArray(List, Value, Name, Found)
  TYPE(ValueList_t), POINTER : List
  CHARACTER(LEN=*) :: Name
  LOGICAL, OPTIONAL :: Found
  REAL(KIND=dp), POINTER :: Value(:, :)
```

### 3.1 Reading constant scalars

Solver Input File:

```
Constants
  MyConstant = Real 123.456
End
```

Code (ElmerProgramming/case1/MyLibrary.f90):

```
SUBROUTINE MySolver(Model, Solver, dt, Transient)
  Use DefUtils
  IMPLICIT NONE
  TYPE(Solver_t) :: Solver
  TYPE(Model_t) :: Model
  REAL(KIND=dp) :: dt
  LOGICAL :: Transient

  ! Read constant scalar from Constants-block:
  !-----
  REAL(KIND=dp) :: MyConstant
  LOGICAL :: Found

  MyConstant = GetConstReal(Model % Constants, "MyConstant", Found)
  IF(.NOT.Found) CALL Fatal("MySolver", "Unable to find MyConstant")
  PRINT *, "MyConstant =", MyConstant

END SUBROUTINE MySolver
```

Output:

```
MyConstant = 123.45600000
```

### 3.2 Reading constant vectors

Solver Input File:

```
Solver 1
  MyVector(3) = Real 1.2 3.4 5.6
End
```



## Code (ElmerProgramming/case2/MyLibrary.f90)

```

SUBROUTINE MySolver(Model, Solver, dt, Transient)
  Use DefUtils
  IMPLICIT NONE
  TYPE(Solver_t) :: Solver
  TYPE(Model_t)  :: Model
  REAL(KIND=dp)  :: dt
  LOGICAL :: Transient

  ! Read constant vector from Solver-block:
  !-----
  REAL(KIND=dp), POINTER :: MyVector(:, :)
  LOGICAL :: Found

  CALL GetConstRealArray(Solver % Values, MyVector, "MyVector", Found)
  IF(.NOT.Found) CALL Fatal("MySolver", "Unable to find MyVector")
  PRINT *, "MyVector =", MyVector(:,1)

END SUBROUTINE MySolver

```

## Output:

```

MyVector =   1.20000000000    3.40000000000    5.60000000000

```

## 3.3 Reading constant matrices

## Solver Input File:

```

Material 1
  MyMatrix(2,3) = Real 11 12 13 \
                  21 22 23
End

```

## Code (ElmerProgramming/case3/MyLibrary.f90):

```

SUBROUTINE MySolver(Model, Solver, dt, Transient)
  Use DefUtils
  IMPLICIT NONE
  TYPE(Solver_t) :: Solver
  TYPE(Model_t)  :: Model
  REAL(KIND=dp)  :: dt
  LOGICAL :: Transient

  ! Read constant matrix from Material-block
  !-----
  REAL(KIND=dp), POINTER :: MyMatrix(:, :)
  LOGICAL :: Found
  TYPE(ValueList_t), POINTER :: Material

  Material => Model % Materials(1) % Values
  CALL GetConstRealArray(Material, MyMatrix, "MyMatrix", Found)
  IF(.NOT.Found) CALL Fatal("MySolver", "Unable to find MyMatrix")
  PRINT *, "Size of MyMatrix =", SIZE(MyMatrix,1), "x", SIZE(MyMatrix,2)
  PRINT *, "MyMatrix(1,:) =", MyMatrix(1,:)
  PRINT *, "MyMatrix(2,:) =", MyMatrix(2,:)

END SUBROUTINE MySolver

```

## Output:

```

Size of MyMatrix =          2 x          3
MyMatrix(1,:) =   11.000000000    12.000000000    13.000000000
MyMatrix(2,:) =   21.000000000    22.000000000    23.000000000

```

## 3.4 Exercises

You can access your global solution vector in your subroutine as follows:

```

TYPE(Variable_t), POINTER :: MyVariable
REAL(KIND=dp), POINTER :: MyVector(:)
INTEGER, POINTER :: MyPermutation(:)
...
MyVariable => Solver % Variable
MyVector => MyVariable % Values
MyPermutation => MyVariable % Perm

```

In the case of a scalar field, you can then set the value of the field e.g. in node 3 as

```
MyVector(MyPermutation(3)) = 123.456
```

The vector `MyPermutation` is related to band width optimization and it is always on by default. You can turn the optimization off by adding the line `Bandwidth optimization = FALSE` in the Solver-block of your SIF. In this case the permutation vector `MyPermutation` becomes the identity map.

Write a user defined subroutine that loops over the elements, reads scalar field data from the Body Force-block of the SIF, and copies the nodal data into the global solution vector (that is, "solve" the equation  $u = f$ ). Use the following Body Force block:

```
Body Force 1
  MyForce = Variable Coordinate 1
  Real
    -1.0 0.0
    1.0 123.456
  End
End
```

Visualize the solution with ElmerPost. The solution should grow linearly from left to right.

## Chapter 4

# Partial Differential Equations

### 4.1 Model problem

In this section, we will consider the boundary value problem

$$\begin{aligned} -\Delta u &= f & \text{in } \Omega, \\ u &= 0 & \text{on } \partial\Omega, \end{aligned}$$

where  $\Omega \subset R^d$  is a smooth bounded domain ( $d = 1, 2, 3$ ) and  $f = 1$ .

The problem can be written as

$$\frac{1}{2} \int_{\Omega} |\nabla u|^2 d\Omega - \int_{\Omega} f u d\Omega = \min!$$

where the minimum is taken over all sufficiently smooth functions that satisfy the kinematical boundary conditions on  $\partial\Omega$ .

### 4.2 FEM

The Galerkin FEM for the problem is obtained by dividing  $\Omega$  into finite elements and by introducing a set of mesh dependent basis functions  $\{\phi_1, \phi_2, \dots, \phi_n\}$ . The approximate solution is written as a linear combination of the basis and determined from the condition that it minimizes the energy:

$$u_n = \sum_{i=1}^n \phi_i u_i \quad (u_i \in R)$$

and

$$\frac{1}{2} \int_{\Omega} |\nabla u_n|^2 d\Omega - \int_{\Omega} f u_n d\Omega = \min!$$

The solution satisfies

$$\sum_{j=1}^n A_{ij} u_j = f_i, \quad i = 1, 2, \dots, n,$$

with

$$A_{ij} = \int_{\Omega} \nabla \phi_i \cdot \nabla \phi_j d\Omega$$

and

$$f_i = \int_{\Omega} f \phi_i d\Omega.$$

In practice, the coefficients  $A_{ij}$  are computed by summing over the elements:

$$A_{ij} = \sum_E A_{ij}^E$$

with

$$A_{ij}^E = \int_E \nabla \phi_i \cdot \nabla \phi_j \, d\Omega$$

The integrals over the elements are evaluated through a mapping  $f_E : \hat{E} \rightarrow E$ , where  $\hat{E}$  is a fixed reference element:

$$A_{ij}^E = \int_{\hat{E}} \nabla \phi_i \cdot \nabla \phi_j |J_E| \, d\hat{\Omega}$$

where  $|J_E|$  is the determinant of the Jacobian matrix of  $f_E$ . In most cases,  $f_E$  is either an affine or an isoparametric map from the unit triangle, square, tetrahedron, hexahedron etc., into the actual element.

Finally, the integral over the reference element is computed numerically with an appropriate quadrature. Elmer uses the Gauss-quadrature by default, as most of the FE-codes:

$$A_{ij}^E = \sum_{k=1}^N \nabla \phi_i(\xi_k) \cdot \nabla \phi_j(\xi_k) w_k |J_E(\xi_k)|$$

where  $\xi_k$  is the integration point and  $w_k$  is the integration weight.

So, the system matrices and vectors of the FEM are formed by implementing a loop over the elements, by computing the local matrices and vectors with an appropriate quadrature, and by assembling the global system from the local contributions.

### 4.3 Implementation

Let us next implement the method in Elmer and write a user defined subroutine for the Poisson equation. To begin with, let us allocate memory for the local matrices and vectors. This is done once and for all in the beginning of the subroutine:

```
INTEGER :: N
TYPE(Mesh_t), POINTER :: Mesh
LOGICAL :: AllocationsDone = .FALSE.
REAL(KIND=dp), ALLOCATABLE :: Matrix(:,,:), Vector(:)
SAVE AllocationsDone, LocalMatrix, LocalVector

IF(.NOT.AllocationsDone) THEN
  Mesh => GetMesh(Solver)
  N = Mesh % MaxElementNodes
  ALLOCATE(Matrix(N,N))
  ALLOCATE(Vector(N))
END IF
```

The next step is to implement a loop over all active elements, call a subroutine that computes the local matrices and vectors (to be specified later), and assemble the global system by using the DefUtils subroutine `DefaultUpdateEquations()`:

```
INTEGER :: i
TYPE(Element_t), POINTER :: Element

DO i = 1, GetNOFActive(Solver)
  Element => GetActiveElement(i)
  N = GetElementNOFNodes(Element)
  CALL ComputeLocal(Element, N, Matrix, Vector)
  CALL DefaultUpdateEquations(Matrix, Vector, Element)
END DO
```

The assembly is finalized by calling the DefUtils subroutine `DefaultFinishAssembly()`. Dirichlet boundary conditions are set by calling the subroutine `DefaultDirichletBCs()`. The final algebraic system is solved by the DefUtils function `DefaultSolve()`:

```
REAL(KIND=dp) :: Norm

CALL DefaultFinishAssembly(Solver)
CALL DefaultDirichletBCs(Solver)
Norm = DefaultSolve(Solver)
```

It remains to implement the subroutine `ComputeLocal()` that makes the local computations. We will contain this subroutine in the main subroutine to simplify things:

```
SUBROUTINE MySolver(Model, Solver, dt, Transient)
...

CONTAINS

SUBROUTINE ComputeLocal(Element, N, Matrix, Vector)
  TYPE(Element_t), POINTER :: Element
  INTEGER :: N
  REAL(KIND=dp) :: Matrix(:, :)
  REAL(KIND=dp) :: Vector(:)
  ...
END SUBROUTINE ComputeLocal

END SUBROUTINE MySolver
```

The first thing to do in `ComputeLocal()` is to clear the matrix and vector:

```
Matrix = 0.0d0
Vector = 0.0d0
```

Next, we will get information about the node points:

```
TYPE(Nodes_t) :: Nodes
SAVE Nodes

Matrix = 0.0d0
Vector = 0.0d0

CALL GetElementNodes(Nodes, Element)
```

The Gauss points for our element are obtained by calling `GaussPoints()`

```
TYPE(GaussIntegrationPoints_t) :: IP

IP = GaussPoints(Element)
```

The local matrix and vector are integrated numerically by implementing a loop over the Gauss points, by evaluating the nodal basis functions in these points, and by computing the inner products:

```
INTEGER :: i
REAL(KIND=dp) :: detJ, Basis(N), dBasisdx(N,3)
LOGICAL :: stat

DO i = 1, IP % n
  stat = ElementInfo(Element, Nodes, &
    IP % u(i), IP % v(i), IP % w(i), &
    detJ, Basis, dBasisdx)
END DO
```

In this loop, we will finally compute the inner products of the basis and their gradients, multiply the result by the weight of the Gauss point, and by the determinant of the Jacobian matrix of the mapping from the reference element:

```
Matrix(1:N, 1:N) = Matrix(1:N, 1:N) + &
  MATMUL(dBasisdx, TRANSPOSE(dBasisdx)) * IP % s(i) * detJ

Vector(1:N) = Vector(1:N) + Basis * IP % s(i) * detJ
```

The implementation is now complete.

Let us finally test the method by creating a finite element mesh e.g. with ElmerGrid or ElmerGUI (1, 2, and 3d are all fine), and by using the following SIF:

```
Header
  Mesh DB "." "."
End

Simulation
  Simulation Type = Steady state
  Steady State Max Iterations = 1
  Post File = case.ep
End

Body 1
  Equation = 1
End
```

```
Equation 1
  Active Solvers(1) = 1
End

Solver 1
  Equation = "MyEquation"
  Procedure = "MyLibrary" "MySolver"
  Variable = -dofs 1 "MyScalar"
End

Boundary condition 1
  Target boundaries(1) = 1
  MyScalar = Real 0
End
```